INTEG3 – 3-point Gaussian Integration

© 1980 Valentín Albillo

Abstract

INTEG3 is a program written in 1980 for the SHARP PC-1211 pocket computer and compatibles to evaluate the definite integral between given limits of an arbitrary user-supplied function f(x) using the 3-point Gauss-Legendre quadrature formula applied over a number of subintervals. Two worked examples are given.

Keywords: *definite integration, 3-point Gauss-Legendre quadrature, SHARP pocket computers, PC-1210, PC-1211, PC-1212*

1. Introduction

INTEG3 is a simple 5-line BASIC program that I wrote in 1980 for the *SHARP PC-1211* pocket computer and compatible models (runs *as-is* in the *PC-1210* and *PC-1212* and with minimal or no changes in other models), which evaluates the definite integral between specified limits of some user-supplied function *f(x)* using the *3-point Gauss-Legendre* quadrature formula applied over a given number of subintervals. The method is as follows: we have

$$I = \int_a^b f(x)\,.\,dx$$

but first of all the change of variable $x = \frac{1}{2}(b + a) + \frac{1}{2}(b - a)t$, $dx = \frac{1}{2}(b - a).\,dt$ transforms the interval *(a, b)* into the interval *(-1, 1)*. The *3-point Gauss-Legendre* quadrature formula then gives:

$$\int_{-1}^{1} f(x)\,.\,dx = \frac{8}{9}f(0) + \frac{5}{9}\ (f\left(\sqrt{\frac{3}{5}}\right) + f\left(-\sqrt{\frac{3}{5}}\right))$$

which is *exact* for polynomial *f(x)* up to the 5th degree and a 5th-order approximation otherwise. See ***Notes***.

2. Program Listing

```
10: "A" USING: INPUT "A=";A,"B=";B,"SUBS=";N: H=(B-A)/N,B=A,G=0,K= √.6
20: FOR Z=1 TO N: A=B,B=B+H
30: X=0: GOSUB 50: F=8Y,X=K: GOSUB 50: F=F+5Y,X=-K: GOSUB 50: F=(F+5Y)*(B-A)/18
40: G=G+F: NEXT Z: BEEP 1: PRINT "INTEGRAL=";G
50: X=((B+A)+(B-A)*X)/2
```

3. Usage Instructions

See the worked examples to understand how to use the program.

4. Examples

The following examples can be used to check that the program was correctly entered and to understand its usage.

4.1 Example 1

Evaluate the integral $I = \int_{3.59}^{20.19} x^5\,.\,dx$

In **PRO** Mode, enter the following program line to define the function *f(x)* to be integrated:

```
60: Y=X^5: RETURN
```

In **DEF** Mode, proceed as follows to evaluate the integral using just one (sub)interval:

```
        SHFT  A    →  A=_
3.59    ENTER      →  B=_
20.19   ENTER      →  SUBS=_
1       ENTER      →  INTEGRAL= 11288934.08
```

The theoretical value is: $(20.19^6 - 3.59^6)/6 =$ ***11288934.08***92297... so the computed value has all 10 digits correct. The reason for such accuracy despite the large integration interval and using just one (sub)interval is because the *3-point Gauss-Legendre* quadrature formula is *exact* for polynomial functions up to the 5th degree, which x^5 is.

4.2 Example 2

Evaluate the integral $I = \int_0^1 \frac{\text{Sin}(x)}{x}.dx$

In **PRO** Mode, enter the following program line to define the function *f(x)* to be integrated:

```
60: Y=SIN(X)/X: RETURN
```

In **DEF** Mode, proceed as follows to evaluate the integral using *1*, *2* and *4* subinterval(s):

```
        SHFT  A    →  A=_
0       ENTER      →  B=_
1       ENTER      →  SUBS=_
1       ENTER      →  INTEGRAL= 9.460831344E-01   (computed using 1 (sub)interval)
    ... repeat the first 3 lines above ...
2       ENTER      →  INTEGRAL= 9.460830715E-01   (computed using 2 subintervals)
    ... repeat the first 3 lines above ...
4       ENTER      →  INTEGRAL= 9.460830704E-01   (computed using 4 subintervals)
```

The theoretical value rounds to ***0.9460830704*** so the computed values have *6*, *8* and *10* correct digits for *1*, *2* and *4* subintervals, respectively.

Notes

*1.*The *3-point Gauss-Legendre* quadrature formula gives 5th-order accuracy for arbitrary *f(x)* using just 3 evaluations per subinterval. This is much better than *Simpson's Rule*, which only gives 3rd-order accuracy with the same number of evaluations.

2. Integrals which have *singularities* at one or both endpoints can aalso be computed as *f(x)* is not evaluated there.

References

Francis Scheid (1988). *Schaum's Outline of Theory and Problems of Numerical Analysis, 2nd Edition*.

Copyrights